Define the photon wave function ψ_n as

$$\psi_n = \exp(in(kx - \omega t))$$

where n is the number of photons, ω is the angular frequency, and k is the wave number. The angular frequency and wave number are related to frequency f as follows.

$$\omega = 2\pi f$$
$$k = \omega / c$$

Note that ψ_n has a normalization problem which will be ignored for this demo.

$$\int_{-\infty}^{+\infty} \psi_n^* \psi_n \, dx = \int_{-\infty}^{+\infty} dx = \infty$$

The energy operator is

$$\hat{E} = i\hbar \frac{\partial}{\partial t}$$

The eigenvalues of the energy operator are $\hbar\omega$ times the number of photons.

$$\hat{E}\psi_n = \hbar\omega n \psi_n$$

The annihilation operator a subtracts a photon.

$$a\psi_n = \psi_1^* \left(\psi_n \frac{\hat{E}\psi_n}{\hbar\omega} \right)^{1/2} = \sqrt{n}\, \psi_{n-1}$$

The creation operator $a^\dagger$ adds a photon.

$$a^\dagger \psi_n = \psi_1 \left(\psi_n \frac{\hat{E}\psi_n}{\hbar\omega} + \psi_n^2 \right)^{1/2} = \sqrt{n+1}\, \psi_{n+1}$$

The number operator counts the number of photons.

$$N\psi_n = a^\dagger a \psi_n = n\psi_n$$

Note that

$$\hat{E}\psi_n = \hbar\omega N \psi_n$$

The following script defines ψ_n and checks the operators.

```
-- www.eigenmath.org/photon-wave-function.txt
k = omega / c
psi(n) = exp(i * n * (k*x - omega*t))
psi0 = psi(0)
psi1 = psi(1)
psi2 = psi(2)
psi3 = psi(3)
psi4 = psi(4)
-- define operators
phat(psi) = -i * hbar * d(psi,x)
Ehat(psi) = i * hbar * d(psi,t)
a(psi) = conj(psi1) * sqrt(psi * Ehat(psi)/hbar/omega)
adag(psi) = psi1 * sqrt(psi * Ehat(psi)/hbar/omega + psi^2)
N(psi) = adag(a(psi))
"check lowering operator (0 = ok)"
a(psi1) - sqrt(1) * psi0
a(psi2) - sqrt(2) * psi1
a(psi3) - sqrt(3) * psi2
a(psi4) - sqrt(4) * psi3
"check raising operator (0 = ok)"
adag(psi0) - sqrt(1) * psi1
adag(psi1) - sqrt(2) * psi2
adag(psi2) - sqrt(3) * psi3
adag(psi3) - sqrt(4) * psi4
"check number operator (0 = ok)"
N(psi0) - 0 * psi0
N(psi1) - 1 * psi1
N(psi2) - 2 * psi2
N(psi3) - 3 * psi3
N(psi4) - 4 * psi4
```